Работа с файлами в операционной системе MS DOS


В ранних версиях DOS для обработки файлов использовались так называемые блоки управления файлами – FCB (File Control Block). Использование блоков управления файлами вызывало определённые сложности и неудобства. Например, нельзя было создать файл с указанием путей – можно было только указывать имя файла.


В DOS, начиная с версии 2.0, появились новые расширенные функции для работы с файлами, упрощающие выполнение стандартных операций. Для новых версий DOS рекомендуется использовать именно эти функции.

Данные в формате ASCIIZ

При использовании многих расширенных функций для дисковых операций необходимо указать адрес строки в формате ASCIIZ, содержащей пути и имя файла. Строка в формате ASCIIZ – это последовательность символов, завершающаяся двоичным нулём, например:

FILENAME DB ‘c:\autoexec.bat’,0

Если строка не содержит путей, то подразумевается файл в текущей директории. Для расширенных функций адрес ASCIIZ-строки указывается в регистровой паре DS:DX – в регистре DS указывается сегмент строки, а в регистре DX – её смещение.

Файловый номер и коды возврата


Функции создания и открытия файлов после своего выполнения устанавливают в регистре AX число, представляющее собой файловый номер. Файловый номер используется другими функциями (например, функцией чтения или записи файла) для идентификации открытого или созданного файла. Стандартные устройства ввода-вывода не нуждаются в операции открытия и могут использовать непосредственные файловые номера: 0 – ввод, 1 – вывод, 2 – вывод сообщений об ошибках, 3 – внешнее устройство, 4 – принтр. Таким образом, функцию записи файла, например, можно использовать для вывода информации на экран или на принтер.


После выполнения любой файловой операции флаг CF сбрасывается, если функция выполнилась успешно и устанавливается, если в ходе выполнения функции возникла какая-либо ошибка. При этом в регистр заносится код ошибки, зависящий от операции (коды ошибок приведены в лекции 10).

Функции DOS для работы с файлами


После выполнения каждой из расмотренных функций DOS для работы с файлами рекомендуется проверить значение флага CF для определения успешности выполнения функции. При неуспешном выполнении функции следует завершить программу и выдать сообщение о соответствующей ошибке.

Функция 3CH


Назначение: создание нового файла или перезапись старого файла. Имя создаваемого файла указывается в виде ASCIIZ-строки в регистровой паре DS:DX. В регистр CX заносится необходимый атрибут файла (см. табл. 1). В случае успешного выполнения функции создаётся новый элемент оглавления (файл) с заданным атрибутом, а в регистре AX возвращается файловый номер созданного файла. Если файл уже существует, то его длина устанавливается равной 0 для перезаписи.

Таблица 1 – коды атрибутов файлов.

Код атрибута
Атрибут
Код атрибута
Атрибут

00H
Нормальный файл
08H
Метка диска

01H
Файл только для чтения
10H
Директория

02H
Скрытый файл
20H
Архивный файл

04H
Системный файл




Пример использования функции 3CH:

FileName DB ‘newfile.txt’,0
; имя файла

Handle DW ?

. . .

MOV AH, 3CH

MOV CX, 0




; создание обычного файла

MOV DX, OFFSET FileName

; ASCIIZ-строка

INT 21H

JC Error

MOV Handle, AX



; сохранение файлового номера

. . .

Error:

. . .




; произошла ошибка


При задании атрибутов файла можно комбинировать значения из таблицы 1 операцией ИЛИ. При этом файл может иметь несколько атрибутов. Например, если в регистр CX занести значение 03H, то файл будет иметь атрибуты «только для чтения» и «скрытый». При перезаписи существующего файла его атрибуты не изменяются. Файл с атрибутом «нормальный» имеет атрибут «архивный», поэтому при любом значении регистра CX файл будет иметь атрибут «архивный». Хотя регистр CX может иметь значение 10H, создать директорию при помощи функции 3CH не удастся. Директорию можно создать специально предназначенной для этого функцией 39H (см. далее). При попытке перезаписать существующий файл с атрибутом «только чтение» функция установит флаг CF, свидетельствующий о возникновении ошибки, и занесёт в регистр AX код ошибки 05H (доступ запрещён).

Функция 3DH


Назначение: открытие существующего файла. Эта функция должна вызываться перед тем, как будет вызвана функция чтения файла. Функция 3DH проверяет правильность имени файла и его наличие на диске. При вызове функции регистровая пара DS:DX должна содержать адрес ASCIIZ-строки, а регистр AL код доступа:

0 – открыть файл только для чтения;

1 – открыть файл только для записи;

2 – открыть файл для чтения и записи.

Код доступа имеет значение при использовании DOS в сетевом окружении или если установлена программа SHARE.EXE. При попытке открыть файл, доступный только для чтения, с кодом доступа 1 или 2 функция установит флаг CF, свидетельствующий о возникновении ошибки, и занесёт в регистр AX код ошибки 05H (доступ запрещён). Следует отметить, что при открытии файла для записи используется функция создания файла 3CH. Если файл нужно открыть и для чтения и для записи, то нужно использовать функцию 3DH. Если файл с указанным именем не существует, то функция 3DH вернёт код ошибки 02H (файл не найден) или 03H (путь не найден) в регистре AX.


Пример использования функции 3DH:

FileName DB ‘newfile.txt’,0
; имя файла

Handle DW ?

. . .

MOV AH, 3DH

MOV AL, 0




; открытие файла для чтения

MOV DX, OFFSET FileName

; ASCIIZ-строка

INT 21H

JC Error

MOV Handle, AX



; сохранение файлового номера

. . .

Error:

. . .




; произошла ошибка

Функция 3FH


Назначение: чтение данных из открытого файла. Для выполнения функции в регистр BX необходимо поместить файловый номер, в регистр CX – число байт, которое необходимо прочитать, а в регистровую пару DS:DX – адрес области памяти, куда будут прочитаны данные. Перед чтением файл должен быть открыт функцией 3DH. При успешном выполнении функции в регистр AX заносится число реально прочитанных байт. Если число реально прочитанных байт меньше, чем было задано, то достигнут конец файла.


Так как DOS ограничивает число одновременно открытых файлов (параметр FILES в CONFIG.SYS), то программа, успешно обработавшая какой-либо файл, должна закрыть его.


Пример использования функции 3FH:

Buf DB 512 DUP (?)

; буфер для чтения

Handle DW ?


; файловый номер

. . .

MOV AH, 3FH

MOV BX, Handle

MOV CX, 512


; чтение 512 байт

MOV DX, OFFSET Buf

; настройка DX на буфер Buf

INT 21H

JC Error

CMP AX, 512

JNE EndOfFile

. . . 



; функция выполнилась успешно

Error:

. . .



; возникла ошибка

EndOfFile:

. . .



; достигнут конец файла

В случае ошибки в регистре AX устанавливается код 05H (нет доступа) или 06H (неправильный файловый номер).

Функция 40H


Назначение: запись данных в открытый файл. Для выполнения функции в регистр BX необходимо поместить файловый номер, в регистр CX – число байт, которое необходимо записать, а в регистровую пару DS:DX – адрес области памяти, откуда будут записаны данные. Перед записью файл должен быть открыт для записи или создан функцией 3CH. При успешном выполнении функции в регистр AX заносится число реально записанных байт. Если число реально записанных байт меньше, чем было задано, то это означает, что диск переполнен.


Пример использования функции 40H:

Buf DB 512 DUP (?)

; буфер для записи

Handle DW ?


; файловый номер

. . .

MOV AH, 40H

MOV BX, Handle

MOV CX, 512


; запись 512 байт

MOV DX, OFFSET Buf

; настройка DX на буфер Buf

INT 21H

JC Error

CMP AX, 512

JNE DiskFull

. . . 



; функция выполнилась успешно

Error:

. . .



; возникла ошибка

DiskFull:

. . .



; диск переполнен

В случае ошибки в регистре AX устанавливается код 05H (нет доступа) или 06H (неправильный файловый номер).

Функция 3EH


Назначение: закрытие ранее открытого файла. Функция закрывает файл, открытый для чтения или для записи. Если закрывается файл, открытый для записи, функция 3EH записывает оставшиеся данные из файлового буфера на диск, корректирует оглавление и таблицу FAT. Для выполнения функции в регистр BX необходимо занести файловый номер.


Пример использования функции 3EH:

Handle DW ?


; файловый номер

. . .

MOV AH, 3EH

MOV BX, Handle

INT 21H

В случае ошибки в регистре AX устанавливается код 06H (неправильный файловый номер).

Функция 41H


Назначение: удаление указанного файла. Для выполнения функции в регистровой паре DS:DX необходимо указать адрес ASCIIZ-строки, содержащей имя файла. Функцией 41H нельзя удалить файл с атрибутом «только для чтения».


Пример использования функции 41H:

FileName DB ‘c:\docs\my.txt’,0
; имя файла

. . .

MOV AH, 41H

MOV DX, OFFSET FileName


; ASCIIZ-строка
INT 21H

В случае ошибки в регистре AX устанавливается код 02H (файл не найден) или 05H (нет доступа).

Функция 42H


Назначение: управление файловым указателем. Файловый указатель – 32-битное это число, которое определяет текущую позицию чтения/записи для файла. При открытии файла файловый указатель устанавливается в 0 и увеличивается при последующих операциях чтения/записи (последовательный доступ к файлу). Для получения прямого доступа к записям файла можно изменять значение файлового указателя при помощи функции 42H.


Для установки файлового указателя необходимо поместить в регистр BX файловый номер, а в регистровую пару CX:DX требуемое значение файлового указателя (смещение). В регистре DX устанавливаются младшие 16 бит смещения, а в регистре CX – старшие 16 бит. В регистре AL должен быть установлен один из кодов, которые определяют точку отсчёта смещения:

0 – смещение от начала файла;

1 – смещение от текущего значения файлового указателя, которое может быть в любом месте, включая начало файла;

2 – смещение от конца файла. Размер файла можно определить, установив регистровую пару CX:DX в 0 и используя код 2 в регистре AL.


Пример использования функции 42H:

Handle DW ?

; файловый номер

. . .

MOV AH, 42H

MOV AL, 0

MOV BX, Handle

MOV CX, 0010H

; смещение 1052672 от начала файла

MOV DX, 1000H

INT 21H

JC Error

. . .


; функция выполнилась успешно

Error:

. . .


; возникла ошибка

В случае ошибки в регистре AX устанавливается код 01H (ошибка кода отсчёта) или 06H (неверный файловый номер).

Функция 43H


Назначение: проверка или изменение атрибута файла. Эта функция позволяет получить атрибут файла или изменить его. Для получения атрибута регистр AL должен содержать 0. Для изменения атрибута регистр AL должен содержать 1, а регистр CX – новое значение атрибуты (см. табл. 1). Если регистр AL содержит 0, то после выполнения функции в регистр CX будет помещён атрибут файла. Регистровая пара DS:DX содержит адрес ASCIIZ-строки с именем файла.


Пример использования функции 43H:

FileName DB ‘c:\docs\my.txt’,0
; имя файла

. . .

MOV AH, 43H

MOV AL, 01H



; установка


MOV CX, 01H



; атрибута «только для чтения»

MOV DX, OFFSET FileName

; ASCIIZ-строка

INT 21H


При использовании функции 43H атрибут «нормальный» означает отсутствие всех атрибутов файла. С помощью функции нельзя изменением атрибута из файла получить директорию и наоборот

В случае ошибки в регистре AX устанавливается код 02H, 03H или 05H.

Функция 56H


Назначение: переименование существующего файла. Регистровая пара DS:DX должна содержать адрес ASCIIZ-строки с именем переименовываемого файла, а регистровая пара ES:DI – адрес ASCIIZ-строки с новым значением имени файла. Буквы дисков вобеих ASCIIZ-строках должны быть одинаковыми, а пути могут различаться. Таким образом, функция 56H позволяет не только переименовывать файл, но и переносить его из одного каталога в другой.


Пример использования функции 56H:

FileName1 DB ‘c:\docs\my.txt’,0
; старое имя файла

FileName2 DB ‘c:\temp\old.txt’,0
; новое имя файла

. . .

MOV AH, 56H

MOV AX, SEG FileName1

MOV DS, AX




; настройка DS

MOV ES, AX




; настройка ES

MOV DX, OFFSET FileName1


; ASCIIZ-строка

MOV DI, OFFSET FileName2


; ASCIIZ-строка
INT 21H

В случае ошибки в регистре AX устанавливается код 03H (путь не найден), 05H (доступ запрещён) или 17H (разные дисководы).


Следующий пример демонстрирует программу, которая создаёт в текущем каталоге файл ENV.TXT и записывает в него своё окружение (переменные окружения и их значения), при этом строки разделяются символами перевода строки (в блоке переменных окружения программы строки разделяются нулями).

CODE SEGMENT

        ASSUME CS:CODE, DS:CODE
; CS и DS указывают на сегмент CODE

        ORG 100H



; COM программа

Start:  MOV AH, 3CH


; функция создания файла

        MOV CX, 0



; с нормальным атрибутом

        MOV DX, OFFSET FileName
; ASCIIZ-строка

        INT 21H



; вызов DOS

        JC Error



; переход по ошибке

        MOV BX, AX


; сохранение файлоаого номера в BX

        MOV AX, DS:2CH


; настройка DS

        MOV DS, AX


; на сегмент окружения программы

        XOR SI, SI


; обнуление SI

M2:     MOV DX, SI


; запоминание начала строки в DX

M1:     LODSB



; поиск нуля

        CMP AL, 0



; (конца строки)

        JNZ M1

        CMP BYTE PTR [SI], 0

; если это последняя строка,

        JZ Exit



; то выход из программы

        MOV CX, SI


; вычисление

        SUB CX, DX


; длины строки

        DEC CX

        MOV AH, 40H


; функция записи в файл

        INT 21H



; вызов DOS

        JC Error



; переход по ошибке
        MOV CX, 2



; запись 2-х байт

        PUSH DS



; сохранение DS

        PUSH CS



; настройка DS

        POP DS



; на сегмент команд (данных)

        MOV DX, OFFSET CrLf

; смещение области для записи

        MOV AH, 40H


; функция записи в файл

        INT 21H



; вызов DOS

        POP DS



; восстановление DS

        JC Error



; переход по ошибке
        JMP M2



; повтор цикла записи строки в файл

Exit:   MOV AH, 3EH


; функция закрытия файла

        INT 21H



; вызов DOS

Error:  RET




; выход из программы

        FileName DB 'ENV.TXT',0
; имя создаваемого файла

        CrLf DB 0Dh,0Ah


; символы перевода строки

CODE ENDS

        END Start



; определение точки входа


Следующий пример демонстрирует программу, выводящую на экран содержимое текстового файла, имя которого указывается в командной строке.

BUF_SIZE EQU 512



; размер буфера для чтения

CODE SEGMENT

        ASSUME CS:CODE, DS:CODE
; CS и DS указывают на сегмент CODE

        ORG 100H



; COM программа

Start:  MOV SI, 80H


; в SI – смещение командной строки

        LODSB



; загрузка в AL длины строки

        CMP AL, 0



; если строка нулевой длины,

        JZ Exit



; то выход из программы

M1:     LODSB



; пропуск пробелов

        CMP AL, ' '


; в начале

        JE M1



; строки

        MOV DX, SI


; настройка DX на 

        DEC DX



; начало имени файла

M2:     LODSB



; поиск признака

        CMP AL, 0DH


; конца

        JNE M2



; строки

        MOV BYTE PTR [SI - 1], 0
; создание ASCIIZ-строки

        MOV AH, 3DH


; функция открытия файла

        MOV AL, 0



; код доступа – только чтение

        INT 21H



; вызов DOS

        JC Error



; переход по ошибке

        MOV BX, AX


; сохранения файлового номера в BX

        MOV CX, BUF_SIZE

; запись в CX размера буфера для чтения

M3:     MOV AH, 3FH


; функция чтения из файла

        MOV DX, OFFSET Buf

; смещение буфера для чтения

        INT 21H



; вызов DOS

        JC Error



; переход по ошибке

        MOV SI, AX


; сохранение числа прочитанных байт
        MOV BYTE PTR [SI + Buf], '$'; создание строки для функции 09H

        MOV AH, 09H


; функция вывода строки на экран

        INT 21H



; вызов DOS

        CMP SI, BUF_SIZE

; если прочитано 512 байт,

        JE M3



; то это ещё не конец файла

Exit:   MOV AH, 3EH


; функция закрытия файла

        INT 21H



; вызов DOS

Error:  RET




; выход из программы

        Buf DB BUF_SIZE+1 DUP (?)
; буфер для чтения

CODE ENDS

        END Start



; точка входа

Функции для работы с каталогами


Функции для работы с каталогами выполняют такие операции, как создание и у даление каталогов, а также получение и установка текущего каталога.

Функция 39H


Назначение: создание каталога. Действие функции аналогично команде DOS «MKDIR». Для выполнения функции в регистровую пару DS:DX необходимо занести адрес ASCIIZ-строки с именем создаваемого каталога. ASCIIZ-строка может включать существующий путь. Функцией 39H можно создать каталог только в уже существующем каталоге, т. е. эта функция не может создать сразу несколько несуществующих каталогов. ASCIIZ-строка для функции 39H может иметь максимум 64 символа (не включая признак конца строки). Если в пути не указана буква диска, то каталог создаётся на текущем диске. То же относится и к пути каталога.


Пример использования функции 39H:

DirName DB ‘c:\games’,0

; имя создаваемого каталога

. . .

MOV AH, 39H

MOV DX, OFFSET DirName

; ASCIIZ-строка

INT 21H

Функция 3AH


Назначение: удаление пустого каталога. Действие функции аналогично команде DOS «RMDIR». Для выполнения функции в регистровую пару DS:DX необходимо занести адрес ASCIIZ-строки с именем удаляемого каталога. ASCIIZ-строка для функции 3AH может иметь максимум 64 символа (не включая признак конца строки). Удаляемый каталог также не должен быть текущим и для не должны быть активными такие программы DOS как JOIN и SUBST.


Пример использования функции 3AH:

DirName DB ‘c:\games’,0

; имя удаляемого каталога

. . .

MOV AH, 3AH

MOV DX, OFFSET DirName

; ASCIIZ-строка

INT 21H

Функция 47H


Назначение: получение текущего каталога для указанного или текущего диска. Для выполнения функции регистровая пара DS:SI должна содержать адрес области памяти, куда функция 47H занесёт значение текущего каталога в формате ASCIIZ. Регистр DL должен содержать номер диска: 0 – текущий, 1 – A, 2 – B и т. д. Область памяти, куда заносится значение текущего каталога, должна быть достаточно большой, чтобы содержать пути максимальной длины.


Пример использования функции 47H:

CurDir DB 256 DUP (?)

; область памяти

. . .

MOV AH, 47H

MOV DL, 3




; диск ‘C:’

MOV SI, OFFSET DirName

; адрес области памяти
INT 21H

Функция 3BH


Назначение: установка текущего каталога. Действие функции аналогично команде DOS «CHDIR». Для выполнения функции регистровая пара DS:SI должна содержать адрес ASCIIZ-строки, определяющей значение текущего каталога. ASCIIZ-строка также может содержать букву диска, для которого устанавливается текущий каталог. Если буква диска не указана, используется текущий диск. Максимальный размер ASCIIZ-строки – 64 символа, не включая признак конца строки.

Пример использования функции 3BH:

CurDir DB ‘c:\games’,0

; значение текущего каталога

. . .

MOV AH, 3BH

MOV DX, OFFSET DirName

; ASCIIZ-строка

INT 21H

Поиск файлов по шаблону


DOS включает в себя две функции, с помощью которых можно искать файлы по заданному шаблону. Шаблон представляет собой ASCIIZ-строку, которая может содержать букву диска, путь и шаблон имени файла. Шаблон имени файла – такой же, как и для команды DOS «DIR».

Функция 4EH


Назначение: поиск первого файла по шаблону. Для выполнения этой функции в регистровой паре DS:DX необходимо указать адрес ASCIIZ-строки, содержащей шаблон, а в регистре CX – атрибут, с которым будут искаться файлы (см. табл. 1). Также для выполнения функции 4EH нужно установить адрес так называемой области передачи данных. Применительно к функции 4EH эта область содержит значения, необходимые для дальнейшего поиска файлов. Для того, чтобы установить адрес области передачи данных необходимо в регистровую пару DS:DX занести требуемый адрес, и вызвать функцию 1AH прерывания DOS 21H. В качестве области передачи данных можно использовать область памяти, отведённую под командную строку (она начинается по смещению 0080H от начала префикса программного сегмента). Таким образом, для установки адреса области передачи данных в регистр DS можно занести сегмент префикса программного сегмента, а в регистр DX – смещение 0080H, а затем вызвать функцию 1AH перрывания DOS 21H.


Если файл по указанному шаблону не найден, функция 4EH устанавливает флаг CF и записывает в регистр AX значение кода ошибки 18 (нет больше файлов). Если файл найден, то область передачи данных заполняется структурой, значения полей которой представлены в таблице 2.

Таблица 2 – поля структуры области передачи данных.

Смещение

(десятичное)
Размер поля

(в байтах)
Описание

0
21
Зарезервировано DOS для последующего поиска

21
1
Атрибут файла

22
2
Время файла

24
2
Дата файла

26
4
Размер файла

30
13
Имя и расширение в виде ASCIIZ-строки

Поля этой структуры можно использовать для получения информации о найденном файле. Для проведения аналогии, можно отметить, что функция 4EH выполняет такие же действия, что и процедура FindFirst библиотеки Dos языка Паскаль. Функция 4EH обычно всегда используется совместно с функцией 4FH, как будет рассмотрено далее.


Пример использования функции 4EH (для программы типа COM):

Template DB ‘c:\games\*.*’,0
; шаблон для поиска

. . .

MOV AX, CS

; настройка DS

MOV DS, AX

; на префикс программного сегмента

MOV DX, 0080H

; настройка DX на область передачи данных

INT 1AH


; установка текущей области передачи данных

MOV AH, 4EH

; функция поиска первого файла

MOV DX, OFFSET Template

MOV CX, 0


; нормальный атрибут

INT 21H

Функция 4FH


Назначение: поиск следующего файла по шаблону. Для выполнения этой функции регистровая пара DS:DX должна содержать адрес области передачи данных, заполненной функцией 4EH. Если следующий файл по шаблону не найден, функция 4FH устанавливает флаг CF и записывает в регистр AX значение кода ошибки 18 (нет больше файлов). Если файл найден, то область передачи данных заполняется новыми данными о найденном файле.


Пример использования функции 4FH (для программы типа COM):

MOV AX, CS

; настройка DS

MOV DS, AX

; на префикс программного сегмента

MOV DX, 0080H

; настройка DX на область передачи данных

MOV AH, 4FH

; поиск следующего файла по шаблону

INT 21H

Функция 4FH выполняет такие же действия, что и процедура FindNext библиотеки Dos языка Паскаль.
Совместное использование функций 4EH и 4FH


Процесс поиска файлов по шаблону с использованием функций 4EH и 4FH включает в себя следующие шаги:

1. установка адреса области передачи данных функцией 1AH;
2. поиск первого файла по шаблону функцией 4EH;
3. если в регистре AX – код ошибки 18 (нет больше файлов), то переход на шаг 8.

4. установка в регистровой паре DS:DX адреса области передачи данных для функции 4FH;
5. поиск следующего файла по шаблону функцией 4FH;
6. если в регистре AX – код ошибки 18 (нет больше файлов), то переход на шаг 8.

7. повтор поиска – переход на шаг 5.

8. завершение поиска файлов.


При помощи функций 4EH и 4FH можно искать не только файлы, но и каталоги, а также метки дисков – для этого нужно установить в регистре CX для функции 4EH необходимые значения атрибута. Чтобы найти файлы с любыми атрибутами в регистре CX нужно установить все биты атрибутов файлов.


Следующий пример демонстрирует программу, которая выводит на экран имена всех файлов и каталогов из текущего каталога и указывает тип найденного элемента оглавления (файл или каталог).

FileInfo STRUC



; структура области передачи данных

        DB 21 DUP (?)


; резерв DOS

        Attr DB ?



; атрибут файла

        Time DW ?



; время файла 

        Date DW ?



; дата файла 

        FSize DD ?


; размер файла 

        FName DB 13 DUP (?)

; имя файла 

FileInfo ENDS

CODE SEGMENT

        ASSUME CS:CODE, DS:CODE
; CS и DS указывают на сегмент CODE
        DB 80H DUP (?)


; пропуск 80H байт

        Info FileInfo <>

; переменная типа FileInfo

        ORG 100H



; программа типа COM

Start:  MOV DX, 0080H


; в DX – смещение области передачи данных

        INT 1AH



; установка адреса области передачи

        MOV AH, 4EH


; функция поиска первого файла

        MOV DX, OFFSET Template
; смещение шаблона

        MOV CX, 00110111B

; атрибуты – все файлы и каталоги

        INT 21H



; вызов DOS

        JC Exit



; переход, если файлы не найдены

M1:     CALL OutString


; вывод имени файла на экран

        MOV DX, 0080H


; в DX – смещение области передачи данных

        MOV AH, 4FH


; функция поиска следующего файла

        INT 21H



; вызов DOS 

        JNC M1



; повтор поиска, если есть ещё файлы

Exit:   RET




; выход из программы

OutString PROC



; процедура вывода имени файла

        MOV SI, OFFSET Info.Fname
; в SI – смещение имени файла

        MOV DX, SI


; сохранение его в DX

M2:     MOV DI, SI


; в DI – предыдущее значение SI

        LODSB



; поиск нуля

        CMP AL, 0



; в ASCIIZ-строке

        JNZ M2

        MOV AL, ' '


; символ для заполнения

M3:     STOSB



; заполнение ASCIIZ-строки

        CMP DI, OFFSET Info.FName + 13
; если не достигнут конец строки,

        JB M3



; то заполнение следующей позиции

        MOV BYTE PTR [DI], '$'
; формирование строки для функции 09H

        MOV AH, 09H


; функция вывода строки

        INT 21H



; вызов DOS

        MOV CX, 10


; вывод 10 пробелов

        MOV AH, 02H


; функция вывода символа

        MOV DL, ' '


; символ пробела

M4:     INT 21H



; вызов DOS

        LOOP M4



; цикл по CX

        TEST Info.Attr, 10H

; проверка на каталог

        MOV DX, OFFSET File

; в DX смещение строки File

        JZ M5





        MOV DX, OFFSET Subdir

; в DX смещение строки Subdir

M5:     MOV AH, 09H


; функция вывода строки

        INT 21H



; вызов DOS

        MOV AH, 02H


; функция вывода символа
        MOV DL, 0DH


; символ перевода строки

        INT 21H



; вызов DOS 

        MOV DL, 0AH


; символ перевода курсора в начало строки

        INT 21H



; вызов DOS

        RET




; возврат из процедуры

OutString ENDP

        Template DB '*.*',0

; шаблон для поиска файлов

        File DB '<File>  $'



        Subdir DB '<SubDir>$'

CODE ENDS

        END Start



; точка входа

В предложенном примере программы следует обратить внимание на расположение в программе переменной Info – она располагается перед директивой ORG. Если бы поля структуры были определены, то компоновщик выдал бы ошибку, так как COM программа должна начинаться с команды. Но определение переменной Info не определяет данные, так как переменная остаётся неинициализированной. Такой способ расположения переменных (перед директивой ORG) можно использовать, чтобы далее в программе обращаться к полям префикса программного сегмента. Переменная Info, например, настроена на начало командной строки, которая используется под область передачи данных. Определение DB 80H DUP (?) нужно для настройки переменной Info на смещение 0080H в префиксе программного сегмента.

